

Uvjeti i grananje u Small Basicu

Primjer:

Ako na graju školske godine ostvarim odličan uspjeh (odgovor DA), idem na nagradno putovanje. Ako je odgovor negativan (odgovor NE), ostat ću kod kuće

- Uvjet je odličan uspjeh
 - Ako je uvjet ispunjen, taj odgovor smatramo potvrđnim (DA)
 - Ako uvjet nije ispunjen, taj je odgovor niječan (NE)

Small Basic za rješavanje ovakvih problema koristi naredbu IF(ako)...THEN(tada)...ELSE(inače)

Radi bolje preglednosti programa, retke iza naredbi THEN i ELSE uvlačimo tipkom TAB

Primjer: Napišimo algoritam i program koji će provjeriti je li dva broja **a** i **b** jednaka ili ne.

► Rješenje:

Algoritam: **upiši a,b**

ako je a=b onda ispiši 'Brojevi su jednaki.'

inače ispiši 'Brojevi nisu jednaki.'

Program:

```
a = TextWindow.Read()
b = TextWindow.Read()
If (a=b) Then
    TextWindow.WriteLine("Brojevi su jednaki.")
Else
    TextWindow.WriteLine("Brojevi nisu jednaki.")
EndIf
```

Primjer : Napišimo program koji uspoređuje dva broja **a** i **b**.

```
TextWindow.Write("a = ")
a = TextWindow.Read()
TextWindow.Write("b = ")
b = TextWindow.Read()
If (a>b) Then
    TextWindow.WriteLine(a+" > "+b)
Else
    If (a<b) Then
        TextWindow.WriteLine(a+" < "+b)
    Else
        TextWindow.WriteLine(a+" = "+b)
    EndIf
EndIf
```

Zadaci

1. Napraviti program koji će prije 12.00 sati prikazati tekst "Dobro jutro", a nakon 12.00 sati tekst "Dobar dan".

```
If (Clock.Hour < 12) Then
    TextWindow.WriteLine("Dobro jutro ")
Else
    TextWindow.WriteLine("Dobar dan ")
EndIf
```

- U Small Basicu objekt Clock omogućuje pristup datumu i vremenu. Nudi i brojna svojstva pomoću kojih možete dobiti zasebne podatke o danu, mjesecu, godini, satu, minutama i sekundama.

2. Napišimo program koji utvrđuje je li uneseni broj paran ili neparan.

```
TextWindow.Write("Unesi broj ")
broj = TextWindow.ReadNumber()
dijeljenje = Math.Remainder(broj, 2)
If (dijeljenje = 0) Then
    TextWindow.WriteLine("Broj je paran")
Else
    TextWindow.WriteLine("Broj je neparan")
EndIf
```

3. Napišimo program koji za dva zadana broja n i k piše poruku ako je broj n djeljiv brojem k.

```
TextWindow.Write("Upiši prvi broj: ")
n = TextWindow.ReadNumber()
TextWindow.Write("Upiši drugi broj: ")
k = TextWindow.ReadNumber()
If Math.Remainder(n,k) =0 Then
    TextWindow.WriteLine(n + " je djeljiv s " + k )
EndIf
```

- operacija **Math.Remainder** dijeli prvi broj drugim, a zatim vraća ostatak

Pomoću iskaza **Goto** računalu možete narediti da neku radnju ponovi određeni broj puta.

4. Napisat program koji će ispisati prvih 100 brojeva jedan pokraj drugog.

```
TextWindow.WriteLine("Ispis prvih sto brojeva na ekranu!")
TextWindow.WriteLine("")
a=1
start:
TextWindow.Write(a+" ")
a=a+1
If a<=100 Then
  Goto start
Else
  TextWindow.WriteLine("")
EndIf
```

5. Napiši program koji zbraja prvih 100 prirodnih brojeva.

```
TextWindow.WriteLine("Zbroj prvih 100 prirodnih brojeva!")
TextWindow.WriteLine("")
x=1
zbroj=0
pocetak:
If x>100 Then
  Goto kraj
Else
  zbroj=zbroj+x
  x=x+1
  Goto pocetak
EndIf
kraj:
TextWindow.WriteLine("Zbroj prvih 100 brojeva je: "+zbroj)
```

6. Napišimo program koji ispisuje brojeve manje od 25.

```
i = 1
start:
TextWindow.Write(i + " ")
i=i+1
If i<=25 Then
  Goto start
EndIf
```

7. Ispiši sve dvoznamenkaste brojeve koji završavaju znamenkom 8.

```
x=8
start:
TextWindow.Write(x + " ")
x = x+10
If x<100 Then
  Goto start
EndIf
```

Petlje bez logičkog uvjeta

U ovim petljama uvjet je zadan unaprijed, nema potrebe za logičkim ispitivanjem da li je neki uvjet ispunjen ili ne

Naše osmogodišnje školovanje možemo prikazati petljom bez logičkog uvjeta.

Unaprijed znamo da trebamo završiti osam razreda osnovne škole

Kad završimo osam razreda, možemo upisati srednju školu.

U programima to bi značilo da će se neka radnja ponavljati unaprijed zadani broj puta.

Iskaz **For..EndFor** u programerskom se žargonu zove petlja. Pomoću petlje varijabli možete dodijeliti početnu i završnu vrijednost te postići da računalo povećava vrijednost varijable umjesto vas. Svaki put kada poveća vrijednost varijable, računalo izvršava iskaze između riječi **For** i riječi **EndFor**.

- Petlja počinje linijom **FOR ponovi = 1 TO 8**, gdje su **FOR** i **TO** ključne riječi, a **ponovi** je varijabla petlje, (u ovom slučaju **početna vrijednost** je 1, a **završna vrijednost** je 8).

Zadaci:

1. Napisati program koji ispisuje niz brojeva od 1 do 8.

```
For i =1 To 8
  TextWindow.Write(i + " ")
EndFor
```

2. Napisati program koji ispisuje niz brojeva od 8 prema 1.

Ovdje uvodimo dodatnu naredbu **STEP** (korak)

Dio **Step 2** iskaza **For** naređuje računalu da vrijednost varijable **i** povećava za 2 umjesto za uobičajenu vrijednost 1. Pomoću riječi **Step** možete odrediti bilo koji prirast. Čak možete navesti negativnu vrijednost za korak ako želite da računalo broji unatrag

Ako je korak=1, nije potrebno navoditi STEP korak

Ako je korak>0 mora biti početna vrijednost<završna vrijednost

Ako je korak<0 mora biti početna vrijednost>završne vrijednosti

```
For i = 8 To 1 Step -1
  TextWindow.Write(i + " ")
EndFor
```

3. Ispiši sve brojeve od 1 do 10 osim broja 7.

```
For i =1 To 10
  If i<>7 Then
    TextWindow.WriteLine( i + " ")
  EndIf
EndFor
```